3.5.4 Separation of Variables Solutions in Cylindrical Coordinates

The separation of variables process can be applied in cylindrical coordinates. The solution for a cylinder exposed to a step-change in ambient conditions that is presented in Section 3.5.2 was derived using separation of variables in cylindrical coordinates. In this section, the techniques required to solve this type of problem are presented.

Separation of variables for transient problems in cylindrical coordinates follows from our previous discussion of 2-D steady problems in cylindrical coordinates in Section 2.3.3. The governing equation and boundary conditions are derived and the solution is assumed to be the product of a function of radial position (TR, a function of r) and a function of time (Tt, a function of t). With some algebra, this assumption will lead to a second order ordinary differential equation in r and a first order ordinary differential equation in t. The character of the ordinary differential equations must be selected so that the solution in r provides the eigenfunctions; these will be oscillatory Bessel functions rather than modified Bessel functions. The Bessel functions, when weighted by r, are orthogonal and this property can be exploited in order to evaluate the constants of the series solution that satisfy the initial condition.

The technique is illustrated in the context of the problem shown in Figure 3-27. The beverage dispensers that are commonly used to dispense soda and other soft-drinks at restaurants and convenience stores are often ice-cooled. A cylinder (r_{out} = 0.75 cm, r_{in} = 0.25 cm) of stainless steel (k = 15 W/m-K, ρ = 8000 kg/m^3, c = 475 J/kg-K) is placed in the bottom of a bin that is filled with ice. The melting ice will cool the external surface of the cylinder and, eventually, the cylinder will be cooled to a spatially uniform temperature of T_{ini} = 0°C. This process requires a relatively long time (minutes) because the thermal communication between the external surface of the cylinder and the ice is not very good. Fortunately, in most convenient stores there is a substantial amount of time between customers dispensing the same beverage. When the customer activates the dispenser (at time t=0), liquid at T_∞ = 20°C flows through the center of the cylinder. The heat transfer coefficient between the fluid and the internal surface of the cylinder is large and therefore the dispense process can be approximated as a step change of the internal surface temperature of the cylinder. The beverage must be dispensed in less than t_{disp} = 5.0 s (the customer doesn't want to very wait long to fill a cup).



Figure 3-27: A cylinder used in a beverage dispenser.

The beverage dispenser works as a thermal storage unit. The cylinder is cooled to the ice temperature between dispense processes, storing "cold" or cooling potential that is transferred very quickly to the beverage during each dispense process. The relatively large tube wall thickness is necessary so that the tube has a sufficient thermal mass. Because the dispense process is fast, the external surface of the tube is approximately adiabatic during the dispense process. Axial variations in the temperature of the cylinder are neglected so that the problem is 1-D and transient. However, the geometry and boundary conditions differ from those of the cylinder shown in Figure 3-19(b) so that the Heisler chart function (cylinder_T) is not applicable.

The known information is entered in EES:

```
$UnitSystem SI MASS  RAD PA K J
$Tabstops 0.2 0.4 0.6 0.8 3.5

"Input Information"
r_out=0.75 [cm]*convert(cm,m)                          "outer radius"
r_in=0.25 [cm]*convert(cm,m)                           "inner radius"
k=15 [W/m-K]                                           "thermal conductivity"
rho=8000 [kg/m^3]                                      "density"
c=475 [J/kg-K]                                         "specific heat capacity"
T_ini=converttemp(C,K,0 [C])                           "initial temperature"
T_infinity=converttemp(C,K,20 [C])                     "fluid temperature"
L=1 [m]                                                "per unit length of tube"
alpha=k/(rho*c)                                        "thermal diffusivity"
```

The governing differential equation is derived using a differential control volume in r:

$$\dot{q}_r = \dot{q}_{r+dr} + \frac{\partial U}{\partial t} \tag{3-335}$$

or

$$0 = \frac{\partial \dot{q}_r}{\partial r} dr + \frac{\partial U}{\partial t} \tag{3-336}$$

The rate of conductive heat transfer is:

$$\dot{q}_r = -2\,\pi\, r\, L\, k \frac{\partial T}{\partial r} \tag{3-337}$$

and the rate of energy storage is:

$$\frac{\partial U}{\partial t} = 2\,\pi\, r\, L\, dr\, \rho\, c \frac{\partial T}{\partial t} \tag{3-338}$$

Substituting Eqs. (3-337) and (3-338) into Eq. (3-336) leads to:

$$0 = \frac{\partial}{\partial r}\left[-2\,\pi\,r\,L\,k\,\frac{\partial T}{\partial r}\right]dr + 2\,\pi\,r\,L\,dr\,\rho\,c\,\frac{\partial T}{\partial t} \tag{3-339}$$

or

$$\frac{\alpha}{r}\frac{\partial}{\partial r}\left[r\,\frac{\partial T}{\partial r}\right] = \frac{\partial T}{\partial t} \tag{3-340}$$

The initial condition is:

$$T_{t=0} = T_{ini} \tag{3-341}$$

and the spatial boundary conditions are:

$$T_{r=r_{in}} = T_{\infty} \tag{3-342}$$

$$\left.\frac{\partial T}{\partial r}\right|_{r=r_{out}} = 0 \tag{3-343}$$

The spatial boundary condition at $r = r_{in}$ is not homogeneous and therefore the problem must be transformed by defining the temperature difference relative to the fluid temperature:

$$\theta = T - T_{\infty} \tag{3-344}$$

The transformed problem statement becomes:

$$\frac{\alpha}{r}\frac{\partial}{\partial r}\left[r\,\frac{\partial \theta}{\partial r}\right] = \frac{\partial \theta}{\partial t} \tag{3-345}$$

$$\theta_{t=0} = T_{ini} - T_{\infty} \tag{3-346}$$

$$\theta_{r=r_{in}} = 0 \tag{3-347}$$

$$\left.\frac{\partial \theta}{\partial r}\right|_{r=r_{out}} = 0 \tag{3-348}$$

The solution is assumed to be the product of two functions, $\theta R(r)$ and $\theta t(t)$:

$$\theta\left(r,t\right) = \theta R\left(r\right)\theta t\left(t\right) \tag{3-349}$$

Equation (3-349) is substituted into Eq. (3-345):

$$\frac{\alpha}{r}\frac{\partial}{\partial r}\left[r\frac{\partial\left(\theta R\,\theta t\right)}{\partial r}\right]=\frac{\partial\left(\theta R\,\theta t\right)}{\partial t} \tag{3-350}$$

or

$$\theta t\,\frac{\alpha}{r}\frac{d}{dr}\left[r\frac{d\theta R}{dr}\right]=\theta R\frac{d\theta t}{dt} \tag{3-351}$$

which is divided by $\theta R\ \theta t$ in order to obtain:

$$\frac{\dfrac{d}{dr}\left[r\dfrac{d\theta R}{dr}\right]}{r\,\theta R}=\frac{\dfrac{d\theta t}{dt}}{\alpha\,\theta t}=-\lambda^2 \tag{3-352}$$

where $-\lambda^2$ is a constant. Equation (3-352) results in two ordinary differential equations:

$$\frac{d}{dr}\left[r\frac{d\theta R}{dr}\right]+\lambda^2\,r\,\theta R=0 \tag{3-353}$$

$$\frac{d\theta t}{dt}+\alpha\,\lambda^2\,\theta t=0 \tag{3-354}$$

The ordinary differential for θR is solved by Bessel functions; the solution can be identified as discussed in Section 1.8:

$$\theta R=C_1\,\mathrm{BesselJ}\left(0,\lambda\,r\right)+C_2\,\mathrm{BesselY}\left(0,\lambda\,r\right) \tag{3-355}$$

or by using Maple:

```
> restart;
> ODEr:=diff(r*diff(TR(r),r),r)+lambda^2*r*TR(r)=0;
```

$$ODEr:=\left(\frac{d}{dr}\mathrm{TR}(r)\right)+r\left(\frac{d^2}{dr^2}\mathrm{TR}(r)\right)+\lambda^2\,r\,\mathrm{TR}(r)=0$$

```
> thetaRs:=dsolve(ODEr);
```

$$thetaRs:=\mathrm{TR}(r)=\_C1\ \mathrm{BesselJ}(0,\lambda\,r)+\_C2\ \mathrm{BesselY}(0,\lambda\,r)$$

The boundary condition at $r = r_{in}$, Eq. (3-347), leads to:

$$\theta R_{r=r_{in}}=C_1\,\mathrm{BesselJ}\left(0,\lambda\,r_{in}\right)+C_2\,\mathrm{BesselY}\left(0,\lambda\,r_{in}\right)=0 \tag{3-356}$$

which requires that:

$$C_2 = -C_1 \frac{\text{BesselJ}\left(0, \lambda\, r_{in}\right)}{\text{BesselY}\left(0, \lambda\, r_{in}\right)} \tag{3-357}$$

Equation (3-357) is substituted into Eq. (3-355):

$$\theta R = C_1 \left[\text{BesselJ}\left(0, \lambda\, r\right) - \frac{\text{BesselJ}\left(0, \lambda\, r_{in}\right)}{\text{BesselY}\left(0, \lambda\, r_{in}\right)} \text{BesselY}\left(0, \lambda\, r\right) \right] \tag{3-358}$$

The boundary condition at $r = r_{out}$, Eq. (3-348), provides the eigencondition for the problem:

$$\left.\frac{d\theta R}{dr}\right|_{r=r_{out}} = C_1 \frac{d}{dr}\left[\text{BesselJ}\left(0, \lambda\, r\right) - \frac{\text{BesselJ}\left(0, \lambda\, r_{in}\right)}{\text{BesselY}\left(0, \lambda\, r_{in}\right)} \text{BesselY}\left(0, \lambda\, r\right) \right]_{r=r_{out}} = 0 \tag{3-359}$$

The differentiation can be carried out using the rules for Bessel functions provided in Section 1.8 or by using Maple:

```
> diff(BesselJ(0,lambda*r),r);
                              -BesselJ(1, λ r) λ

> diff(BesselY(0,lambda*r),r);
                              -BesselY(1, λ r) λ
```

which allows Eq. (3-359) to be written as:

$$-\lambda\, \text{BesselJ}\left(1, \lambda\, r_{out}\right) + \frac{\text{BesselJ}\left(0, \lambda\, r_{in}\right)}{\text{BesselY}\left(0, \lambda\, r_{in}\right)} \lambda\, \text{BesselY}\left(1, \lambda\, r_{out}\right) = 0 \tag{3-360}$$

There are many solutions to Eq. (3-360), each representing an eigenvalue of the solution. Equation (3-360) is rearranged and used to define a residual (*Res*) that must be zero at each eigenvalue:

$$Res = \text{BesselJ}\left(0, \lambda\, r_{in}\right) \text{BesselY}\left[1, \left(\lambda\, r_{in}\right) \frac{r_{out}}{r_{in}}\right] - \text{BesselJ}\left[1, \left(\lambda\, r_{in}\right) \frac{r_{out}}{r_{in}}\right] \text{BesselY}\left(0, \lambda\, r_{in}\right) \tag{3-361}$$

Equation (3-361) is programmed in EES and used to generate Figure 3-28 which illustrates the residual as a function of $\lambda\, r_{in}$.

```
"Identify eigenvalues"
Res=BesselJ(0,lambdar_in)*BesselY(1,lambdar_in*r_out/r_in)-&
          BesselY(0,lambdar_in)*BesselJ(1,lambdar_in*r_out/r_in)
```



Figure 3-28: The residual as a function of $\lambda\, r_{in}$.

The points where the residual intersects $y = 0$ represent the eigenvalues; notice that the eigenvalues occur in regularly defined intervals (every $\pi/2$) and therefore it is possible to use the technique discussed in Section 3.5.3 to identify these eigenvalues automatically using EES. By plotting the residual as a function of the argument of the Bessel function (rather than as a function of λ) it is always possible to locate these intervals and use this technique. If the residual were plotted as a function of λ (rather than $\lambda\, r_{in}$) then the intervals would change based on r_{in} and the technique described here to automatically identify the roots of Eq. (3-361) would not be as convenient.

Arrays are setup in order to provide appropriate upper and lower bounds and guess values for each eigenvalue. These arrays are used to constrain the solution for value of $\lambda\, r_{in}$ (the entries in the array lambdar_in[]) in the Variable Information window.

```
Nterm=10 [-]                                                    "number of terms"
"Identify eigenvalues"
duplicate i=1,Nterm
    lowerbound[i]=(i-1)*pi/2                                    "lower bound"
    upperbound[i]=i*pi/2                                        "upper bound"
    guess[i]=lowerbound[i]+pi/4                                 "guess"
    BesselJ(0,lambdar_in[i])*BesselY(1,lambdar_in[i]*r_out/r_in)-&
        BesselY(0,lambdar_in[i])*BesselJ(1,lambdar_in[i]*r_out/r_in)=0   "eigencondition"
    lambda[i]=lambdar_in[i]/r_in                                "eigenvalue"
end
```

The solution to the ordinary differential equation in time for a given eigenvalue, Eq. (3-354), is:

$$\theta t_i = C_{3,i} \exp\left(-\lambda_i^2\, \alpha\, t\right) \tag{3-362}$$

The product of θR_i and θt_i written for any eigenvalue λ_i represents a solution to the governing equation that satisfies both of the spatial boundary conditions:

$$\theta_i = \theta R_i \, \theta t_i = C_i \left[\text{BesselJ}\left(0, \lambda_i \, r\right) - \frac{\text{BesselJ}\left(0, \lambda_i \, r_{in}\right)}{\text{BesselY}\left(0, \lambda_i \, r_{in}\right)} \text{BesselY}\left(0, \lambda_i \, r\right) \right] \exp\left(-\lambda_i^2 \, \alpha \, t\right) \quad (3\text{-}363)$$

The sum the solutions for each eigenvalue is also a solution to the problem:

$$\theta = \sum_{i=1}^{\infty} C_i \left[\text{BesselJ}\left(0, \beta_i \, r\right) - \frac{\text{BesselJ}\left(0, \beta_i \, r_{in}\right)}{\text{BesselY}\left(0, \beta_i \, r_{in}\right)} \text{BesselY}\left(0, \beta_i \, r\right) \right] \exp\left(-\beta_i^2 \, \alpha \, t\right) \quad (3\text{-}364)$$

The constants are selected in order to satisfy the initial condition, Eq. (3-346):

$$\sum_{i=1}^{\infty} C_i \left[\text{BesselJ}\left(0, \lambda_i \, r\right) - D_i \, \text{BesselY}\left(0, \lambda_i \, r\right) \right] = T_{ini} - T_{\infty} \quad (3\text{-}365)$$

where

$$D_i = \frac{\text{BesselJ}\left(0, \lambda_i \, r_{in}\right)}{\text{BesselY}\left(0, \lambda_i \, r_{in}\right)} \quad (3\text{-}366)$$

Bessel functions are orthogonal when they are multiplied by the weighting function r. Therefore, each side of Eq. (3-365) is multiplied by the product of the j^{th} eigenfunction and r and then integrated between the two homogeneous boundaries, from $r = r_{in}$ to $r = r_{out}$. As usual, only the i^{th} term in the series remains:

$$C_i \int_{r_{in}}^{r_{out}} \left[\text{BesselJ}\left(0, \lambda_i \, r\right) - D_i \, \text{BesselY}\left(0, \lambda_i \, r\right) \right]^2 r \, dr = \left(T_{ini} - T_{\infty}\right) \int_{r_{in}}^{r_{out}} \left[\text{BesselJ}\left(0, \lambda_i \, r\right) - D_i \, \text{BesselY}\left(0, \lambda_i \, r\right) \right] r \, dr \quad (3\text{-}367)$$

The integral on the left side of Eq. (3-367) is expanded:

$$C_i \left[\underbrace{\int_{r_{in}}^{r_{out}} \text{BesselJ}^2\left(0, \lambda_i \, r\right) r \, dr}_{\left(\text{Integral 1}\right)_i} - 2 \, D_i \underbrace{\int_{r_{in}}^{r_{out}} \text{BesselY}\left(0, \lambda_i \, r\right) \text{BesselJ}\left(0, \lambda_i \, r\right) r \, dr}_{\left(\text{Integral 2}\right)_i} + D_i^2 \underbrace{\int_{r_{in}}^{r_{out}} \text{BesselY}^2\left(0, \lambda_i \, r\right) r \, dr}_{\left(\text{Integral 3}\right)_i} \right] = \left(T_{ini} - T_{\infty}\right) \underbrace{\int_{r_{in}}^{r_{out}} \left[\text{BesselJ}\left(0, \lambda_i \, r\right) - D_i \, \text{BesselY}\left(0, \lambda_i \, r\right) \right] r \, dr}_{\left(\text{Integral 4}\right)_i} \quad (3\text{-}368)$$

Equation (3-369) is expressed in terms of the four integrals in Eq. (3-368):

$$C_i = \frac{\left(T_{ini} - T_\infty\right)\left(\text{Integral 4}\right)_i}{\left[\left(\text{Integral 1}\right)_i \text{-}2\, D_i \left(\text{Integral 2}\right)_i + D_i^2 \left(\text{Integral 3}\right)_i\right]} \tag{3-369}$$

The four integrals in Eq. (3-369) are evaluated in Maple:

```
> restart;
> Integral1[i]:=int(BesselJ(0,lambda[i]*r)*BesselJ(0,lambda[i]*r)*r,r=r_in..r_out);
```

$$Integral1_i := -\frac{1}{2} r\_in^2 \,\text{BesselJ}(0, \lambda_i\, r\_in)^2 - \frac{1}{2} r\_in^2 \,\text{BesselJ}(1, \lambda_i\, r\_in)^2 + \frac{1}{2} r\_out^2 \,\text{BesselJ}(0, \lambda_i\, r\_out)^2 + \frac{1}{2} r\_out^2 \,\text{BesselJ}(1, \lambda_i\, r\_out)^2$$

```
> Integral2[i]:=int(BesselY(0,lambda[i]*r)*BesselJ(0,lambda[i]*r)*r,r=r_in..r_out);
```

$$Integral2_i := -\frac{1}{2} r\_in^2 \,\text{BesselY}(0, \lambda_i\, r\_in)\,\text{BesselJ}(0, \lambda_i\, r\_in) - \frac{1}{2} r\_in^2 \,\text{BesselJ}(1, \lambda_i\, r\_in)\,\text{BesselY}(1, \lambda_i\, r\_in) + \frac{1}{2} r\_out^2 \,\text{BesselY}(0, \lambda_i\, r\_out)\,\text{BesselJ}(0, \lambda_i\, r\_out) + \frac{1}{2} r\_out^2 \,\text{BesselJ}(1, \lambda_i\, r\_out)\,\text{BesselY}(1, \lambda_i\, r\_out)$$

```
> Integral3[i]:=int(BesselY(0,lambda[i]*r)*BesselY(0,lambda[i]*r)*r,r=r_in..r_out);
```

$$Integral3_i := -\frac{1}{2} r\_in^2 \,\text{BesselY}(0, \lambda_i\, r\_in)^2 - \frac{1}{2} r\_in^2 \,\text{BesselY}(1, \lambda_i\, r\_in)^2 + \frac{1}{2} r\_out^2 \,\text{BesselY}(0, \lambda_i\, r\_out)^2 + \frac{1}{2} r\_out^2 \,\text{BesselY}(1, \lambda_i\, r\_out)^2$$

```
> Integral4[i]:=int((BesselJ(0,lambda[i]*r)-D[i]*BesselY(0,lambda[i]*r))*r,r=r_in..r_out);
```

$$Integral4_i := -\left(r\_in\,\text{BesselJ}(1, \lambda_i\, r\_in) - \text{D}_i\, r\_in\,\text{BesselY}(1, \lambda_i\, r\_in) - r\_out\,\text{BesselJ}(1, \lambda_i\, r\_out) + \text{D}_i\, r\_out\,\text{BesselY}(1, \lambda_i\, r\_out)\right) \Big/ \lambda_i$$

and copied to EES in order to evaluate the solution constants:

```
"Evaluate the constants"
duplicate i=1,Nterm
    D[i]=BesselJ(0,lambda[i]*r_in)/BesselY(0,lambda[i]*r_in)
    Integral1[i] = -1/2*r_in^2*BesselJ(0,lambda[i]*r_in)^2-1/2*r_in^2*&
        BesselJ(1,lambda[i]*r_in)^2+1/2*r_out^2*BesselJ(0,lambda[i]*r_out)^2&
        +1/2*r_out^2*BesselJ(1,lambda[i]*r_out)^2
    Integral2[i] = -1/2*r_in^2*BesselY(0,lambda[i]*r_in)*BesselJ(0,lambda[i]*r_in)&
        -1/2*r_in^2*BesselJ(1,lambda[i]*r_in)*BesselY(1,lambda[i]*r_in)+1/2*r_out^2&
        *BesselY(0,lambda[i]*r_out)*BesselJ(0,lambda[i]*r_out)&
        +1/2*r_out^2*BesselJ(1,lambda[i]*r_out)*BesselY(1,lambda[i]*r_out)
```

```
    Integral3[i] = -1/2*r_in^2*BesselY(0,lambda[i]*r_in)^2-1/2*r_in^2*&
        BesselY(1,lambda[i]*r_in)^2+1/2*r_out^2*BesselY(0,lambda[i]*r_out)^2&
        +1/2*r_out^2*BesselY(1,lambda[i]*r_out)^2
    Integral4[i] = -(r_in*BesselJ(1,lambda[i]*r_in)-D[i]*r_in*BesselY(1,lambda[i]*r_in)&
        -r_out*BesselJ(1,lambda[i]*r_out)+D[i]*r_out*BesselY(1,lambda[i]*r_out))/lambda[i]
    C[i]=(T_ini-T_infinity)*Integral4[i]/(Integral1[i]-2*D[i]*Integral2[i]+D[i]^2*Integral3[i])
end
```

The Maple and EES code listed above appears complicated; however, the solution process is automatic and it was not necessary to manually enter these equations. Imagine how much more difficult it would have been to carry out the integrations manually, which is the way these problems were solved before computer tools were available.

A dimensionless radial position ($\tilde{r}$) is defined for convenience:

$$\tilde{r} = \frac{(r - r_{in})}{(r_{out} - r_{in})} \tag{3-370}$$

and the solution is obtained at a particular time and position:

```
"Obtain the solution"
r_hat=(r-r_in)/(r_out-r_in)                          "dimensionless position"
r_hat=0.5                                            "dimensionless position for solution"
time=0.5 [s]                                         "enter the time for the solution"
duplicate i=1,Nterm
    theta[i]=C[i]*(BesselJ(0,lambda[i]*r)-D[i]*BesselY(0,lambda[i]*r))*exp(-lambda[i]^2*alpha*time)
end
T=T_infinity+sum(theta[1..Nterm])                    "temperature"
r_cm=r*convert(m,cm)                                 "radius in cm"
T_C=Converttemp(K,C,T)                               "temperature in C"
```

Figure 3-29 illustrates the temperature as a function of the radius for various times during the dispense process.



Figure 3-29: Temperature as a function of position at various times during the dispense process.

The solution should be checked against physical intuition. Figure 3-29 shows that both spatial boundary conditions ($T_{r=r_{in}} = 20$°C and $\left.\frac{\partial T}{\partial r}\right|_{r=r_{out}} = 0$) are satisfied. The temperature disturbance at the inner radius should take on the order of one diffusive time constant to reach the outer radius. The diffusive time constant is approximately:

$$\tau_{diff} = \frac{\left(r_{out} - r_{in}\right)^2}{4\,\alpha} \tag{3-371}$$

```
tau_diff=(r_out-r_in)^2/(4*alpha)          "diffusive time constant"
```

which leads to τ_{diff} = 1.6 s; this result is approximately consistent with Figure 3-29.

The rate of heat transfer from the beverage to the cylinder (per unit length, L = 1 m) as a function of time for a 5.0 s dispense process can be calculated using the model. The rate of cooling, $\dot{q}_{cool}$, is obtained using Fourier's law at the inner surface:

$$\dot{q}_{cool} = -k\,2\,\pi\,r_{in}\,L\left.\frac{\partial T}{\partial r}\right|_{r=r_{in}} \tag{3-372}$$

Substituting Eq. (3-344) into Eq. (3-372) leads to:

$$\dot{q}_{cool} = -k\,2\,\pi\,r_{in}\,L\left.\frac{\partial \theta}{\partial r}\right|_{r=r_{in}} \tag{3-373}$$

Substituting Eq. (3-364) into Eq. (3-373) leads to:

$$\dot{q}_{cool} = -k\,2\,\pi\,r_{in}\,L\frac{\partial}{\partial r}\left.\left[\sum_{i=1}^{\infty} C_i\left[\text{BesselJ}\left(0,\lambda_i\,r\right) - \frac{\text{BesselJ}\left(0,\lambda_i\,r_{in}\right)}{\text{BesselY}\left(0,\lambda_i\,r_{in}\right)}\text{BesselY}\left(0,\lambda_i\,r\right)\right]\exp\left(-\lambda_i^2\,\alpha\,t\right)\right]\right|_{r=r_{in}} \tag{3-374}$$

or:

$$\dot{q}_{cool} = -k\,2\,\pi\,r_{in}\,L\sum_{i=1}^{\infty} C_i\exp\left(-\lambda_i^2\,\alpha\,t\right)\frac{d}{dr}\left[\text{BesselJ}\left(0,\lambda_i\,r\right) - D_i\,\text{BesselY}\left(0,\lambda_i\,r\right)\right]_{r=r_{in}} \tag{3-375}$$

where D_i is obtained from Eq. (3-366). Maple is used to evaluate the derivative in Eq. (3-375):

```
> eval(diff(BesselJ(0,lambda[i]*r)-D[i]*BesselY(0,lambda[i]*r),r),r=r_in);
```

$$-\text{BesselJ}(1, \lambda_i\, r\_in)\, \lambda_i + D_i\, \text{BesselY}(1, \lambda_i\, r\_in)\, \lambda_i$$

The rate of cooling can be calculated in EES:

```
"rate of cooling"
duplicate i=1,Nterm
     q_dot_cool[i]=-k*2*pi*r_in*L*C[i]*exp(-lambda[i]^2*alpha*time)*(-
BesselJ(1,lambda[i]*r_in)*lambda[i]+D[i]*BesselY(1,lambda[i]*r_in)*lambda[i])
end
q_dot_cool_total=sum(q_dot_cool[1..Nterm])
```

Figure 3-30 illustrates the rate of cooling per unit length as a function of time.



Figure 3-30: Rate of cooling per unit length provided to the beverage as a function of time during the dispense process.